


RLC-vaihtovirtapiiri: resonanssi

Olkoon tarkastelun kohteena tavallinen RLC-vaihtovirtapiiri. Piirissä on kolme komponenttia, R ohmin vastus, L henryn induktanssi ja C faradin kapasitanssi.



Piiriin syötettyyn jännitteeseen kohdistuu kolme eri pudotusta,

$$\begin{aligned} E_L &= L \frac{d I(t)}{d t} && \text{käämin yli,} \\ E_R &= R I(t) && \text{vastuksen yli sekä} \\ E_C &= \frac{1}{C} \int I(t) dt && \text{kondensaattorin yli.} \end{aligned}$$

Näiden summasta syntyy piirin Kirchhoffin lain mukainen sähkömotorinen voima $E(t)$. Mikäli piiriin syötetty jännite on sinimuotoista vaihtovirtaa $E(t) = E_0 \sin(\omega t)$, saadaan piirille yhtälö

$$E(t) = L \frac{d^2 I(t)}{d t^2} + R I(t) + \frac{1}{C} \int I(t) dt = E_0 \sin(\omega t).$$

Integraalitermistä päästään eroon derivoimalla yhtälö ajan suhteen:

$$L \frac{d^2 I(t)}{d t^2} + R \frac{d I(t)}{d t} + \frac{1}{C} I(t) = E_0 \omega \cos(\omega t).$$

Tämä on differentiaaliyhtälö pakotetulle värähtelylle, jossa ulkoinen pakottava jännite on sinimuotoinen amplitudina E_0 .

Jos $E(t) = 0$, kyseessä vapaa värähtelypiiri ja differentiaaliyhtälö on homogeeninen.

Jos piirissä ei ole vastusta, ts. $R = 0$, piiri on vaimentamaton. Vastuksen olemassaolo merkitsee, että piirissä on vaimennus.

Tarkastellaan differentiaaliyhtälön ratkaisuja vakioiden L , R , C , E_0 ja ω eri arvoilla.

Aluksi hävitetään mahdollisista aiemmista laskuista jääneet muuttujat:

Remove["Global`*"]

Virtaa kuvaava toisen kertaluvun yhtälö:

$$\text{yhtalo} = L i''[t] + R i'[t] + 1/C i[t] == \omega E0 \cos[\omega t]$$

$$\frac{i[t]}{C} + R i'[t] + L i''[t] == E0 \omega \cos[t \omega]$$

Vaimentamaton tapaus

Vaimentamattomassa tapauksessa piirissä ei ole vastusta ja yhtälö on

$$\text{yhtalo1} = \text{yhtalo} /. R \rightarrow 0$$

$$\frac{i[t]}{C} + L i''[t] == E0 \omega \cos[t \omega]$$

Alkuehtona olkoon, että piirissä ei tapahdu mitään:

$$\text{alkuehto} = \{i[0] == 0, i'[0] == 0\}$$

$$\{i[0] == 0, i'[0] == 0\}$$

Tämän ratkaisu on

$$\text{rtk1} = \text{DSolve}[\{\text{yhtalo1}, \text{alkuehto}\}, i[t], t]$$

$$\left\{ \left\{ i[t] \rightarrow \frac{1}{-1 + C L \omega^2} \left(C E0 \omega \cos\left[\frac{t}{\sqrt{C} \sqrt{L}}\right] - C E0 \omega \cos\left[\frac{t}{\sqrt{C} \sqrt{L}}\right]^2 \cos[t \omega] - C E0 \omega \cos[t \omega] \sin\left[\frac{t}{\sqrt{C} \sqrt{L}}\right]^2 \right) \right\} \right\}$$

Täsmällinen symbolinen tulos on tarpeen jatkon oivalluksia varten, joten sievennetään ratkaisu:

$$\text{ratkaisu1} = i[t] /. \text{First}[\text{rtk1}] // \text{FullSimplify}$$

$$\frac{C E0 \omega \left(\cos\left[\frac{t}{\sqrt{C} \sqrt{L}}\right] - \cos[t \omega] \right)}{-1 + C L \omega^2}$$

Ratkaisu muodostuu kahdesta kosinitermistä: $\cos(t/\sqrt{LC})$ on peräisin homogeeniyhtälön yleisestä ratkaisusta ja $\cos(t\omega)$ epähomogeeniyhtälön yksittäisratkaisusta. Edellinen kuvaa piirin sisäistä värähtelyä, jälkimmäinen ulkoisen jännitteen taajuudella tapahtuvaa värähtelyä.

Ratkaisu tässä muodossa ei kuitenkaan ole pätevä, jos nimittäjä tulee nolllaksi:

$$\text{restaajuus} = \text{Solve}[-1 + C L \omega^2 == 0, \omega]$$

$$\left\{ \left\{ \omega \rightarrow -\frac{1}{\sqrt{C} \sqrt{L}} \right\}, \left\{ \omega \rightarrow \frac{1}{\sqrt{C} \sqrt{L}} \right\} \right\}$$

Nollaksi tuloa vastaavaa pakotteen taajuutta kutsutaan *resonanssitaajuudeksi*. Tällöin piirin sisäinen värähtelytaajuus ja pakotteen taajuus ovat samat. Kyseessä on tilanne, missä epähomogeenisen yhtälön yksittäisratkaisu saakin erilaisen muodon:

```
yhtalo2 = yhtalo1 /. Last[restaaajuus]
```

$$\frac{i[t]}{C} + L i''[t] = \frac{E_0 \cos\left[\frac{t}{\sqrt{C} \sqrt{L}}\right]}{\sqrt{C} \sqrt{L}}$$

```
rtk2 = DSolve[{yhtalo2, alkuehto}, i[t], t]
```

$$\left\{ \left\{ i[t] \rightarrow \frac{1}{4L} \left(-\sqrt{C} E_0 \sqrt{L} \cos\left[\frac{t}{\sqrt{C} \sqrt{L}}\right] + \sqrt{C} E_0 \sqrt{L} \cos\left[\frac{t}{\sqrt{C} \sqrt{L}}\right] \cos\left[\frac{2t}{\sqrt{C} \sqrt{L}}\right] + 2 E_0 t \sin\left[\frac{t}{\sqrt{C} \sqrt{L}}\right] + \sqrt{C} E_0 \sqrt{L} \sin\left[\frac{t}{\sqrt{C} \sqrt{L}}\right] \sin\left[\frac{2t}{\sqrt{C} \sqrt{L}}\right] \right) \right\} \right\}$$

```
ratkaisu2 = i[t] /. First[rtk2] // FullSimplify
```

$$\frac{E_0 t \sin\left[\frac{t}{\sqrt{C} \sqrt{L}}\right]}{2L}$$


Tämä on tulkittavissa värähtelytermiksi, jossa virran amplitudi $E_0 t / (2L)$ kasvaa rajatta muuttujan t mukana. Käytännössä tämä johtaisi piirin palamiseen.

Arvoilla $L = 0.1 \text{ H}$, $C = 0.001 \mu\text{F}$, $E_0 = 230 \text{ V}$ saadaan resonanssitaajuudella seuraava kuvio:

```
numratkaisu2 = ratkaisu2 /. {L -> 0.1, C -> 0.001, E0 -> 230}
```

```
1150. t Sin[100. t]
```

```
Plot[numratkaisu2, {t, 0, 1.5}]
```



- Graphics -

Resonanssitaajuus on

```
1 / Sqrt[L C] /. {L -> 0.1, C -> 0.001}
```


```
100.
```

Jos pakotteen taajuus on lähellä resonanssitaajuutta, saadaan erikoinen värähtely:

```
numratkaisu1 = ratkaisu1 /. {L -> 0.1, C -> 0.001, E0 -> 230, omega -> 90}
```

```
-108.947 (-Cos[90 t] + Cos[100. t])
```

```
Plot[numratkaisu1, {t, 0, 1.5}]
```



- Graphics -

Tässä pakote aluksi vahvistaa piirissä kulkevaa virtaa kuten resonanssitajuuden tapauksessakin, mutta koska taajuudet eivät olekaan täsmälleen samat, värähtelyt siirtyvät vähitellen vastakkaisiin vaiheisiin, ja pakote alkaa sammuttaa piirin virtaa.

Syntynyt värähtely voidaan myös ymmärtää sinivärähtelyksi, jolla on sinimuotoinen vaihteleva amplitudi:

```
Cos[100 t] - Cos[90 t] == - 2 Sin[5 t] Sin[95 t] // Simplify
```

```
True
```

Vaimennettu tapaus

Yleisen vaimennetun tapauksen ratkaisu on

```
yhtalo
```

$$\frac{i[t]}{C} + R i'[t] + L i''[t] == E_0 \omega \cos[t \omega]$$

```
rtk = DSolve[yhtalo, i[t], t]
```

$$\left\{ \left\{ i[t] \rightarrow e^{\frac{(-CR - \sqrt{C} \sqrt{-4L + CR^2}) t}{2CL}} C[1] + e^{\frac{(-CR + \sqrt{C} \sqrt{-4L + CR^2}) t}{2CL}} C[2] - \frac{(4(-CE_0 L^2 \omega \cos[t \omega] + C^2 E_0 L^3 \omega^3 \cos[t \omega] - C^2 E_0 L^2 R \omega^2 \sin[t \omega]))}{((-2L + CR^2 - \sqrt{C} R \sqrt{-4L + CR^2} + 2CL^2 \omega^2)(-2L + CR^2 + \sqrt{C} R \sqrt{-4L + CR^2} + 2CL^2 \omega^2))} \right\} \right\}$$

```
ratkaisu = i[t] /. First[rtk] // Simplify
```

$$e^{-\frac{(R + \frac{\sqrt{-4L + CR^2}}{\sqrt{C}}) t}{2L}} C[1] + e^{-\frac{(R - \frac{\sqrt{-4L + CR^2}}{\sqrt{C}}) t}{2L}} C[2] + \frac{CE_0 \omega ((1 - CL \omega^2) \cos[t \omega] + CR \omega \sin[t \omega])}{1 - 2CL \omega^2 + C^2 \omega^2 (R^2 + L^2 \omega^2)}$$

Ratkaisu koostuu kahdesta termistä:

```
sisainen = Select[ratkaisu, !FreeQ[#, C[1]] &] + Select[ratkaisu, !FreeQ[#, C[2]] &]
```

$$e^{-\frac{\left(R + \frac{\sqrt{-4 L C R^2}}{2 C}\right) t}{2 L}} C[1] + e^{-\frac{\left(R - \frac{\sqrt{-4 L C R^2}}{2 C}\right) t}{2 L}} C[2]$$

```
ulkoinen = ratkaisu /. {C[1] -> 0, C[2] -> 0}
```

$$\frac{C E 0 \omega \left((1 - C L \omega^2) \cos[t \omega] + C R \omega \sin[t \omega] \right)}{1 - 2 C L \omega^2 + C^2 \omega^2 (R^2 + L^2 \omega^2)}$$

Sisäisen värähtelyn termi esittää vaimenevaa värähtelyä, sillä muuttujan t kerroin kummassakin eksponentissa on joko negatiivinen reaali- tai kompleksiluku, jonka reaali-osa on negatiivinen. Nämä termit siis kuolevat vähitellen pois.

Ulkoinen värähtelyn termi voidaan muokata seuraavasti. Pyritään saattamaan se muotoon, jossa on vain yksi sinifunktio kertoimena sopiva *amplitudi* A ja argumentissa sopiva *vaihesiirto* δ .

```
uusimuoto = A Sin[ω t + δ] // TrigExpand
```

$$A \cos[t \omega] \sin[\delta] + A \cos[\delta] \sin[t \omega]$$

Vaativalla että sini- ja kosinitermien kertoimet vanhassa ja uudessa esitysmuodossa ovat samat, saadaan ehdot, joista pyritään ratkaisemaan A ja δ .

```
ehdot = MapThread[Equal, {Coefficient[ulkoinen, {Sin[ω t], Cos[ω t]}],  
  Coefficient[uusimuoto, {Sin[ω t], Cos[ω t]}]}]
```

$$\left\{ \frac{C^2 E 0 R \omega^2}{1 - 2 C L \omega^2 + C^2 \omega^2 (R^2 + L^2 \omega^2)} = A \cos[\delta], \frac{C E 0 \omega (1 - C L \omega^2)}{1 - 2 C L \omega^2 + C^2 \omega^2 (R^2 + L^2 \omega^2)} = A \sin[\delta] \right\}$$

```
ratk = Solve[ehdot, {A, δ}] // FullSimplify // PowerExpand // FullSimplify
```

General::spell1 : Possible spelling error: new symbol name "ratk" is similar to existing symbol "rtk". More...

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. More...

$$\left\{ \left\{ A \rightarrow -\frac{C E 0 \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}, \delta \rightarrow -\text{ArcCos}\left[-\frac{C R \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}\right] \right\}, \right. \\ \left\{ A \rightarrow -\frac{C E 0 \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}, \delta \rightarrow \text{ArcCos}\left[-\frac{C R \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}\right] \right\}, \\ \left\{ A \rightarrow \frac{C E 0 \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}, \delta \rightarrow -\text{ArcCos}\left[\frac{C R \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}\right] \right\}, \\ \left. \left\{ A \rightarrow \frac{C E 0 \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}, \delta \rightarrow \text{ArcCos}\left[\frac{C R \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}\right] \right\} \right\}$$

Pakotteen aiheuttama virta on siis sinimuotoinen amplitudina

```
amplitudi = A /. Last[ratk]
```

$$\frac{C E 0 \omega}{\sqrt{1 + C \omega^2 (C R^2 + L (-2 + C L \omega^2))}}$$

Amplitudi riippuu pakotteen taajuudesta. Pyritään määrittämään maksimi-amplitudi ja vastaava taajuus:

```
derivaatta = D[amplitudi,  $\omega$ ] // Simplify
```

$$\frac{C E_0 - C^3 E_0 L^2 \omega^4}{(1 - 2 C L \omega^2 + C^2 \omega^2 (R^2 + L^2 \omega^2))^{3/2}}$$

```
maksimikohta = Solve[derivaatta == 0,  $\omega$ ]
```

$$\left\{ \left\{ \omega \rightarrow -\frac{1}{\sqrt{C} \sqrt{L}} \right\}, \left\{ \omega \rightarrow -\frac{i}{\sqrt{C} \sqrt{L}} \right\}, \left\{ \omega \rightarrow \frac{i}{\sqrt{C} \sqrt{L}} \right\}, \left\{ \omega \rightarrow \frac{1}{\sqrt{C} \sqrt{L}} \right\} \right\}$$

```
maksimiampitudi = amplitudi /. Last[maksimikohta] // Simplify // PowerExpand
```

$$\frac{E_0}{R}$$

Maksimiarvo saadaan siis resonanssitaajuudella.

Piirretään pakotteen (ulkoisen jännitteen) ja resonanssitaajuutta vastaavan virran kuvaajat:

```
numulkoinen = ulkoinen /. {L -> 0.1, C -> 0.0001, R -> 5.0, E0 -> 230}
```

$$\frac{0.023 \omega ((1 - 0.00001 \omega^2) \cos[t \omega] + 0.0005 \omega \sin[t \omega])}{1 - 0.00002 \omega^2 + 1. \times 10^{-8} \omega^2 (25. + 0.01 \omega^2)}$$


```
jannite = E0 Sin[ $\omega$  t] /. E0 -> 230
```

$$230 \sin[t \omega]$$

```
res = 1/Sqrt[L C] /. {L -> 0.1, C -> 0.0001, R -> 5.0, E0 -> 230}
```

$$316.228$$

```
Plot[Evaluate[{jannite, numulkoinen} /.  $\omega$  -> res],  
{t, 0, 0.1}, PlotStyle -> {RGBColor[0, 0, 0], RGBColor[0, 1, 0]}]
```




- Graphics -

Kuviossa on jännite mustalla ja virta vihreällä.

Vastaava kuvio, kun kyseessä ei ole resonanssitaajuus:

```
Plot[Evaluate[{jannite, numulkoinen} /.  $\omega \rightarrow 280$ ],  
  {t, 0, 0.1}, PlotStyle → {RGBColor[0, 0, 0], RGBColor[0, 1, 0]}]
```



- Graphics -

Tehtäviä

Tutki, miten alkuehtojen muuttaminen vaikuttaa ratkaisuun vaimentamattomassa tapauksessa. Miten ratkaisun luonne muuttuu, kun ollaan resonanssikohdan lähellä?

Piirrä kuvaaja, joka esittää vaimennetun tapauksen virtaa pakotteen taajuuden funktiona. Onko vaimennetun piirin tapauksessa sellainen ratkaisu mahdollinen, jossa virran amplitudi rajatta kasvaa, ts. piiri palaa?

Linkit

[vaihtovirtapiirin pakotettu värähtely](#) (virtap3.nb)
[epähomogeeninen vakiokertoaminen lineaariyhtälö](#)

JP & SKK 08.05.2001